Funktionales Programmieren
(Praktische Informatik 3)

Berthold Hoffmann
Studiengang Informatik
Universität Bremen

Winter 03/04

Vorlesung vom 24.11.2001: Algebraische Datentypen

Inhalt

- Letzte VL:
  Funktionsabstraktion durch Funktionen höherer Ordnung.
- Heute: Datenabstraktion durch algebraische Datentypen.
- Einfache Datentypen: Aufzählungen und Produkte
- Der allgemeine Fall
- Bekannte algebraische Datentypen: `Maybe`, Bäume
- Geordnete Bäume
- Abgeleitete Klasseninstanzen

Nichtrekursive Algebraische Datentypen

Was ist Datenabstraktion?

- Typsynonyme sind keine Datenabstraktion
  - `type Day = Int` wird textuell expandiert.
  - Keinerlei Typsicherheit:
    - `Freitag + 15` ist kein Wochentag;
    - `Freitag * Montag` ist kein Typfehler.
  - Kodierung `0 = Montag` ist willkürlich und nicht eindeutig.
- Deshalb:
  - Wochentage sind nur Montag, Dienstag, . . . , Sonntag.
  - Alle Wochentage sind unterschiedlich.
- $\Longrightarrow$ Einfachster algebraischer Datentyp: Aufzählungstyp.

Aufzählungen

- Beispiel: Wochentage

```
data Weekday = Mo | Tu | We | Th | Fr | Sa | Su
```

- Konstruktoren:

```
Mo :: Weekday, Tu :: Weekday, We :: Weekday, ...
```

  - Diese Konstruktoren sind automatisch mit deklariert.

- Funktionsdefinition durch pattern matching:

```
isWeekend :: Weekday -> Bool
isWeekend Sa = True
isWeekend Su = True
isWeekend _  = False
```

Produkte

- Beispiel: Datum besteht aus Tag, Monat, Jahr:

```
data Date  = Date Day Month Year
type Day   = Int
data Month = Jan | Feb | Mar | Apr | May | Jun
           | Jul | Aug | Sep | Oct | Nov | Dec
type Year  = Int
```

- Beispielwerte:

```
today     = Date 24 Nov 2003
bloomsday = Date 16 Jun 1904
fstday    = Date 1 Jan 1
```

- Konstruktor:

```
Date :: Day-> Month-> Year-> Date
```

- Funktionsdefinition:
  - Über pattern matching Zugriff auf Argumente der Konstruktoren:

```
day  :: Date-> Day
year :: Date-> Year
day  (Date d m y) = d
year (Date d m y) = y
```

  - `day`, `year` sind Selektoren:

```
day  today     = 26
year bloomsday = 1904
```

Alternativen

- Beispiel: Eine geometrische Figur ist
  - eine Kreis, gegeben durch Mittelpunkt und Durchmesser,
  - oder ein Rechteck, gegeben durch zwei Eckpunkte,
  - oder ein Polygon, gegeben durch Liste von Eckpunkten.

Alternativen

- Beispiel: Eine geometrische Figur ist
  - eine Kreis, gegeben durch Mittelpunkt und Durchmesser,
  - oder ein Rechteck, gegeben durch zwei Eckpunkte,
  - oder ein Polygon, gegeben durch Liste von Eckpunkten.

```
type Point = (Double, Double)
data Shape = Circ Point Double
           | Rect Point Point
           | Poly [Point]
```

- Ein Konstruktor pro Variante.

```
Circ :: Point-> Double-> Shape
```

Funktionen auf Geometrischen Figuren

- Funktionsdefinition durch pattern matching.
- Beispiel: Anzahl Eckpunkte.

```
corners :: Shape-> Int
corners (Circ _ _)        = 0
corners (Rect _ _)        = 4
corners (Poly ps)         = length ps
```

  - Konstruktor-Argumente werden in ersten Fällen nicht gebraucht.

- Translation um einen Punkt (Vektor):

```
move :: Shape-> Point-> Shape
move (Circ m d) p    = Circ (add p m) d
move (Rect c1 c2) p = Rect (add p c1) (add p c2)
move (Poly ps) p     = Poly (map (add p) ps)
```

- Translation eines Punktes:
  - Durch Addition

```
add :: Point-> Point-> Point
add (x, y) (u, v) = (x+ u, y+ v)
```

- Berechung der Fläche:
  - Einfach für Kreis und Rechteck.

```
area :: Shape-> Double
area (Circ _ d) = pi* d
area (Rect (x1, y1) (x2, y2)) =
      abs ((x2- x1)* (y2- y1))
```



- Fläche für Polygone:
  - Vereinfachende Annahme: Polygone sind konvex
  - Reduktion auf einfacheres Problem und Rekursion:



```
area (Poly ps) | length ps < 3  = 0
area (Poly (p1:p2:p3:ps)) =
           triArea p1 p2 p3 +
           area (Poly (p1:p3:ps))
```

- Fläche für Dreieck mit den Eckpunkten (x_1, y_1), (x_2, y_1), (x_3, y_3):

$$A = \frac{|(x_2 - x_1)(y_3 - y_1) - (x_3 - x_1)(y_2 - y_1)|}{2}$$

```
triArea :: Point-> Point-> Point-> Double
triArea (x1,y1) (x2,y2) (x3, y3) =
    abs ((x2-x1)*(y3-y1) - (x3-x1)*(y2-y1)) / 2
```

- Distanz zwischen zwei Punkten (Pythagoras):




```
dist :: Point-> Point-> Double
dist (x1, y1) (x2, y2) =
  sqrt((x1-x2)^2+ (y2- y1)^2)
```

Rekursive Algebraische Datentypen

Das allgemeine Format

Definition von $\mathtt{T}$: $\mathtt{data\ T} = \mathtt{C}_1\ \mathtt{t}_{1,1} \ldots \mathtt{t}_{1,k_1}$

$\ldots$

$\mid\ \mathtt{C}_n\ \mathtt{t}_{n,1} \ldots \mathtt{t}_{n,k_n}$

- Konstruktoren (und Typen) werden groß geschrieben.
- Konstruktoren $\mathtt{C}_1, \ldots, \mathtt{C}_n$ sind disjunkt:

$$\mathtt{C}_i\ \mathtt{x}_1 \ldots \mathtt{x}_n = \mathtt{C}_j\ \mathtt{y}_1 \ldots \mathtt{y}_m \Rightarrow i = j$$

- Konstruktoren sind injektiv:

$$\mathtt{C}\ \mathtt{x}_1 \ldots \mathtt{x}_n = \mathtt{C}\ \mathtt{y}_1 \ldots \mathtt{y}_n \Rightarrow \mathtt{x}_i = \mathtt{y}_i$$

- Konstruktoren erzeugen den Datentyp:

$$\forall x \in T.\, x = \mathtt{C}_i\ \mathtt{y}_1 \ldots \mathtt{y}_m$$

Rekursive Datentypen

Der definierte Typ `T` kann rechts benutzt werden.

- Beispiel: einfache arithmetische Ausdrücke sind
  - Zahlen (Literale), oder
  - Addition zweier Ausdrücke, oder
  - Subtraktion zweier Ausdrücke.

```
data Expr = Lit Int
          | Add Expr Expr
          | Sub Expr Expr
```

- Funktionen darauf sind meist auch rekursiv.

- Ausdruck auswerten:

```
eval :: Expr-> Int
eval (Lit n)= n
eval (Add e1 e2) = eval e1+ eval e2
eval (Sub e1 e2) = eval e1- eval e2
```

- Ausdruck ausgeben:

```
print :: Expr-> String
print (Lit n) = show n
print (Add e1 e2) = "("++ print e1++ "+"++ print e
print (Sub e1 e2) = "("++ print e1++ "-"++ print e
```

- Testen.

- Primitive Rekursion auf Ausdrücken
  - Einen Wert für Rekursionsverankerung
  - Eine binäre Funktion für Addition
  - Eine binäre Funktion für Subtraktion

```
foldE :: (Int-> a)-> (a-> a-> a)-> (a-> a-> a)
              -> Expr-> a
foldE b a s (Lit n)     = b n
foldE b a s (Add e1 e2) = a (foldE b a s e1)
                            (foldE b a s e2)
foldE b a s (Sub e1 e2) = s (foldE b a s e1)
                            (foldE b a s e2)
```

- Damit Auswertung und Ausgabe:

```
eval'  = foldE id (+) (-)
print' =
  foldE show
        (\s1 s2-> "("++ s1++ "+"++ s2++ ")")
        (\s1 s2-> "("++ s1++ "-"++ s2++ ")")
```

Polymorphie

- Algebraische Datentypen parametrisiert über Typen:

```
data Pair a = Pair a a
```

- Paar: zwei beliebige Elemente gleichen Typs. Zeigen.

Polymorphie

- Algebraische Datentypen parametrisiert über Typen:

```
data Pair a = Pair a a
```

- Paar: zwei beliebige Elemente gleichen Typs. Zeigen.
- Elemente vertauschen:

```
twist :: Pair a-> Pair a
twist (Pair a b) = Pair b a
```

- Map für Paare:

```
mapP :: (a-> b)-> Pair a-> Pair b
mapP f (Pair a b)= Pair (f a) (f b)
```

Listen!

- Eine Liste von `a` ist
  - entweder leer
  - oder ein Kopf `a` und eine Restliste `List a`

```
data List a = Nil | Cons a (List a)
```

- Syntaktischer Zucker der eingebauten Listen:

```
data [a] = [] | a : [a]
```

  - Kann so nicht definiert werden!
  - Operatoren beginnend mit “`:`”sind als binäre Konstruktoren möglich.

- Funktionsdefinition:

```
fold :: (a-> b-> b)-> b-> List a-> b
fold f e Nil              = e
fold f e (Cons a as) = f a (fold f e as)
```

- Mit `fold` alle primitiv rekursiven Funktionen, wie:

```
map' f    = fold (Cons . f) Nil
length'   = fold ((+).(const 1)) 0
filter' p = fold (\x-> if p x then Cons x
                               else id) Nil
```

  - Konstante Funktion `const:: a-> b-> a` aus dem Prelude.

Modellierung von Fehlern: `Maybe a`

- Typ `a` plus Fehlerelement
  - Im Prelude vordefiniert.

```
data Maybe a = Just a | Nothing
```

- `Nothing` wird im „Fehlerfall“ zurückgegeben. Bsp:

```
find :: (a-> Bool)-> [a]-> Maybe a
find p []     = Nothing
find p (x:xs) = if p x then Just x
                else find p xs
```

Binäre Bäume

- Ein binärer Baum ist
  - Entweder leer,
  - oder ein Knoten mit genau zwei Unterbäumen.
  - Knoten tragen ein Label.

```
data Tree a = Null
            | Node (Tree a) a (Tree a)
```

- Andere Möglichkeiten:
  - Label für Knoten und Blätter:

```
data Tree' a b = Null'
               | Leaf' b
               | Node' (Tree' a b) a (Tree' a b)
```

- Test auf Enthaltensein:

```
member :: Eq a=> Tree a-> a-> Bool
member Null _ = False
member (Node l a r) b =
   a == b || (member l b) || (member r b)
```

- Primitive Rekursion auf Bäumen:
  - Rekursionsschritt:
    - Label des Knoten
    - Zwei Rückgabewerte für linken, rechten Unterbaum
  - Rekursionsanfang

```
foldT :: (a-> b-> b-> b)-> b-> Tree a-> b
foldT f e Null = e
foldT f e (Node l a r) =
   f a (foldT f e l) (foldT f e r)
```

- Damit: Elementtest, `map`

```
member' t x =
   foldT (\e b1 b2-> e == x || b1 || b2) False t

mapT :: (a-> b)-> Tree a-> Tree b
mapT f = foldT (flip Node . f) Null
```

- Testen.

- Traversion:

```
preorder  :: Tree a-> [a]
inorder   :: Tree a-> [a]
postorder :: Tree a-> [a]
preorder   = foldT (\x t1 t2-> [x]++ t1++ t2) []
inorder    = foldT (\x t1 t2-> t1++ [x]++ t2) []
postorder  = foldT (\x t1 t2-> t1++ t2++ [x]) []
```

  - Äquivalente Definition ohne `foldT`:

```
preorder' Null           = []
preorder' (Node l a r)   = [a] ++preorder' l ++preorder' r
```

- Testen.

Geordnete Bäume

- Voraussetzung:
  - Ordnung auf `a` (`Ord a`)
  - Es soll für alle `Node a l r` gelten:

    $$\texttt{member x l} \Rightarrow \texttt{x} < \texttt{a} \wedge \texttt{member x r} \Rightarrow \texttt{a} < \texttt{x}$$

    Jeder Eintrag kommt höchstens einmal vor

- Test auf Enthaltensein vereinfacht:

```
member :: Ord a=> Tree a-> a-> Bool
member Null _ = False
member (Node l a r) b
   | b < a  = member l b
   | a == b = True
   | b > a  = member r b
```

- Ordnungserhaltendes Einfügen

```
insert :: Ord a=> Tree a-> a-> Tree a
insert Null a = Node Null a Null
insert (Node l a r) b
   | b < a  = Node (insert l b) a r
   | b == a = Node l a r
   | b > a  = Node l a (insert r b)
```

- Problem: Erzeugung ungeordneter Bäume möglich.
  - Lösung erfordert Verstecken der Konstruktoren — nächste VL.

- Löschen:

```
delete :: Ord a=> a-> Tree a-> Tree a
delete x Null = Null
delete x (Node l y r)
   | x < y  = Node (delete x l) y r
   | x == y = join l r
   | x > y  = Node l y (delete x r)
```

- `join` fügt zwei Bäume ordnungserhaltend zusammen.

- Beispiel: Gegeben folgender Baum, dann `delete t 9`



- Beispiel: Gegeben folgender Baum, dann `delete t 9`



  - Wurzel wird gelöscht

- Beispiel: Gegeben folgender Baum, dann `delete t 9`



  - Wurzel wird gelöscht
  - `join` muß Bäume ordnungserhaltend zusammenfügen

- `join` fügt zwei Bäume ordnungserhaltend zusammen.



  - Wurzel des neuen Baums: Knoten links unten im rechten Teilbaum (oder Knoten rechts unten im linken Teilbaum)

- Implementation:
  - `splitTree` spaltet Baum in Knoten links unten und Rest.
  - Testen.

```
join ::  Tree a-> Tree a-> Tree a
join xt Null = xt
join xt yt   = Node xt u nu where
  (u, nu) = splitTree yt
  splitTree :: Tree a-> (a, Tree a)
  splitTree (Node Null a t) = (a, t)
  splitTree (Node lt a rt) =
    (u, Node nu a rt) where
      (u, nu) = splitTree lt
```

Abgeleitete Klasseninstanzen

- Wie würde man Gleichheit auf `Shape` definieren?

```
Circ p1 i1 == Circ p2 i2 = p1 == p2 && i1 == i2
Rect p1 q1 == Rect p2 q2 = p1 == p2 && q1 == q2
Poly ps    == Poly qs    = ps == qs
_          == _          = False
```

- Schematisierbar:
  - Gleiche Konstruktoren mit gleichen Argumente gleich,
  - alles andere ungleich.
- Automatisch generiert durch `deriving Eq`
- Ähnlich `deriving (Ord, Show, Read)`

Wohlbekannte vordefinierte Typklassen

- Wahrheitswerte sind ein Aufzählungstyp

```
data Bool = False | True
     deriving (Eq,Ord,Ix,Enum,Read,Show,Bounded)
```

- Rationale Zahlen und Tupel sind Produkttypen

```
data Integral a => Ratio a = a :% a
type Rational = Ratio Integer
data (a,b) = (a,b)
data (a,b,c) = (a,b,c)
...
```

Zusammenfassung

- Algebraische Datentypen erlauben Datenabstraktion durch
  - Trennung zwischen Repräsentation und Semantik und
  - Typsicherheit.
- Algebraischen Datentypen sind frei erzeugt.
- Bekannte algebraische Datentypen:
  - Aufzählungen, Produkte, Varianten;
  - `Maybe a`, Listen, Bäume, Tupel
- Für geordnete Bäume — die nicht frei erzeugt sind: Konstruktoren verstecken — nächste Vorlesung.